// File: /home/keso/projects/crypto/uptown-panda-token/contracts/UptownPandaPresale.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.6.0;

import "@openzeppelin/contracts/math/SafeMath.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router02.sol";
import "./interfaces/IUptownPanda.sol";
import "./interfaces/IUniswapV2Helper.sol";
import "./interfaces/IUptownPandaFarm.sol";

contract UptownPandaPresale is Ownable {
    using SafeMath for uint256;

    event InvestmentSucceeded(address sender, uint256 weiAmount, uint256 upAmount);

    mapping(address => bool) public whitelistAddresses; // all addresses eligible for presale
    mapping(address => uint256) public investments; // total WEI invested per address (1ETH = 1e18WEI)
    uint256 public whitelistAddressesCount; // total count of whitelisted addresses

    address public immutable uptownPandaAddress; // address of $UP token
    address public immutable uniswapRouterAddress; // address of uniswap router
    address public immutable liquidityLockAddress; // address where liquidity pool tokens will be locked for 2 years
    address payable public immutable teamAddress; // address where invested ETH will be transfered to

    uint256 public constant UP_FARM_INITIAL_SUPPLY = 27500 * 1 ether;
    uint256 public constant UP_ETH_FARM_INITIAL_SUPPLY = 55000 * 1 ether;
    uint256 public constant WETH_FARM_INITIAL_SUPPLY = 7500 * 1 ether;
    uint256 public constant WBTC_FARM_INITIAL_SUPPLY = 7500 * 1 ether;

    address public immutable upFarmAddress; // address for farming with $UPs
    address public immutable upEthFarmAddress; // address for farming with $UP-ETH uniswap token
    address public immutable wethFarmAddress; // address for farming with WETH
    address public immutable wbtcFarmAddress; // address for farming with WBTC
    address private immutable wbtcAddress; // address to use for WBTC farm intialization

    uint256 public immutable PRESALE_WEI_HARD_CAP; // max amount of ETH collected in presale (in WEI)
    uint256 public constant PRESALE_PRICE_MULTIPLIER = 3; // how many times more $UP presale investor receives vs listing price
    uint256 public constant INVESTMENT_LIMIT = 2.5 * 1 ether; // 2.5 ETH is maximum investment limit
    uint256 public presaleWeiSupplyLeft; // how many WEI are still available in presale

    bool public isPresaleActive = false; // investing is only allowed if presale is active
    bool public allowWhitelistAddressesOnly = true; // if true only addresses found on whitelist can participate
    bool public wasPresaleEnded = false; // indicates that presale is ended and liqudity is provided

    IUptownPanda private immutable uptownPanda;
    IUniswapV2Router02 private immutable uniswapRouter;

    constructor(
        address _uptownPandaAddress,
        address _uniswapV2HelperAddress,
        address _liquidityLockAddress,
        address _teamAddress,
        address _upFarmAddress,
        address _upEthFarmAddress,
        address _wethFarmAddress,
        address _wbtcFarmAddress,
        address _wbtcAddress,
        uint256 _presaleEthSupply
    ) public {
        uptownPandaAddress = _uptownPandaAddress;
        uptownPanda = IUptownPanda(_uptownPandaAddress);

        address resolvedUniswapRouterAddress = IUniswapV2Helper(_uniswapV2HelperAddress).getRouterAddress();
        uniswapRouterAddress = resolvedUniswapRouterAddress;
        uniswapRouter = IUniswapV2Router02(resolvedUniswapRouterAddress);

        liquidityLockAddress = _liquidityLockAddress;
        teamAddress = payable(_teamAddress);

        upFarmAddress = _upFarmAddress;
        upEthFarmAddress = _upEthFarmAddress;
        wethFarmAddress = _wethFarmAddress;
        wbtcFarmAddress = _wbtcFarmAddress;
        wbtcAddress = _wbtcAddress;

        presaleWeiSupplyLeft = _presaleEthSupply * 1 ether;
        PRESALE_WEI_HARD_CAP = presaleWeiSupplyLeft;
    }

    function addWhitelistAddresses(address[] calldata _whitelistAddresses) external onlyOwner {
        for (uint256 i = 0; i < _whitelistAddresses.length; i++) {
            whitelistAddresses[_whitelistAddresses[i]] = true;
            whitelistAddressesCount = whitelistAddressesCount.add(1); 
        }
    }

    function setIsPresaleActive(bool _isPresaleActive) external onlyOwner {
        if (_isPresaleActive && !uptownPanda.isInitialized()) {
            uptownPanda.initialize(
                address(this),
                uniswapRouter.WETH(),
                upFarmAddress,
                upEthFarmAddress,
                wethFarmAddress,
                wbtcFarmAddress
            );
        }
        isPresaleActive = _isPresaleActive;
    }

    function setAllowWhitelistAddressesOnly(bool _allowWhitelistAddressesOnly) external onlyOwner {
        allowWhitelistAddressesOnly = _allowWhitelistAddressesOnly;
    }

    modifier presaleActive() {
        require(isPresaleActive, "Presale is currently not active.");
        _;
    }

    modifier presaleNotEnded {
        require(!wasPresaleEnded, "Presale was ended.");
        _;
    }

    modifier presaleContractIsMinter {
        require(uptownPanda.getMinter() == address(this), "Presale contract can't mint $UP tokens.");
        _;
    }

    modifier eligibleForPresale() {
        require(!allowWhitelistAddressesOnly || whitelistAddresses[_msgSender()], "Your address is not whitelisted.");
        _;
    }

    modifier presaleSupplyAvailable() {
        require(presaleWeiSupplyLeft > 0, "Presale cap has been reached.");
        _;
    }

    modifier presaleSupplyNotExceeded() {
        require(msg.value <= presaleWeiSupplyLeft, "The amount of ETH sent exceeds the ETH supply left in presale.");
        _;
    }

    receive()
        external
        payable
        presaleActive
        presaleNotEnded
        presaleContractIsMinter
        eligibleForPresale
        presaleSupplyAvailable
        presaleSupplyNotExceeded
    {
        uint256 addressTotalInvestment = investments[_msgSender()].add(msg.value);
        require(addressTotalInvestment <= INVESTMENT_LIMIT, "Max investment per address is 2.5 ETH.");

        uint256 listingPriceMultiplier = uptownPanda.getListingPriceMultiplier();
        uint256 upsToMint = msg.value.mul(listingPriceMultiplier).mul(PRESALE_PRICE_MULTIPLIER);
        uptownPanda.mint(_msgSender(), upsToMint);

        investments[_msgSender()] = addressTotalInvestment;
        presaleWeiSupplyLeft = presaleWeiSupplyLeft.sub(msg.value);

        emit InvestmentSucceeded(_msgSender(), msg.value, upsToMint);
    }

    function endPresale() external onlyOwner presaleNotEnded {
        uint256 listingPriceMultiplier = uptownPanda.getListingPriceMultiplier();
        uint256 liquidityPoolEths = address(this).balance.mul(60).div(100); // 60% goes to liquidity pool FOREVER
        uint256 liquidityPoolUps = liquidityPoolEths.mul(listingPriceMultiplier);

        uptownPanda.mint(address(this), liquidityPoolUps); // mint $UPs for liquidity pool and assign them to presale address
        uptownPanda.approve(address(uniswapRouter), liquidityPoolUps); // approve uniswap router to use $UPs from this address

        address upAddress = address(uptownPanda);
        uint256 transactionDeadline = block.timestamp.add(5 minutes); // transaction should be confirmed in that timeframe
        uniswapRouter.addLiquidityETH{ value: liquidityPoolEths }(
            upAddress,
            liquidityPoolUps,
            liquidityPoolUps,
            liquidityPoolEths,
            liquidityLockAddress,
            transactionDeadline
        );

        // start up farm
        uptownPanda.mint(upFarmAddress, UP_FARM_INITIAL_SUPPLY);
        IUptownPandaFarm(upFarmAddress).startFarming(uptownPandaAddress, uptownPandaAddress, UP_FARM_INITIAL_SUPPLY);

        // start up/eth farm
        uptownPanda.mint(upEthFarmAddress, UP_ETH_FARM_INITIAL_SUPPLY);
        IUptownPandaFarm(upEthFarmAddress).startFarming(
            uptownPandaAddress,
            uptownPanda.uniswapPair(),
            UP_ETH_FARM_INITIAL_SUPPLY
        );

        // start weth farm
        uptownPanda.mint(wethFarmAddress, WETH_FARM_INITIAL_SUPPLY);
        IUptownPandaFarm(wethFarmAddress).startFarming(
            uptownPandaAddress,
            uniswapRouter.WETH(),
            WETH_FARM_INITIAL_SUPPLY
        );

        // start wbtc farm
        uptownPanda.mint(wbtcFarmAddress, WBTC_FARM_INITIAL_SUPPLY);
        IUptownPandaFarm(wbtcFarmAddress).startFarming(uptownPandaAddress, wbtcAddress, WBTC_FARM_INITIAL_SUPPLY);

        teamAddress.transfer(address(this).balance); // remaining ETHs (40%) go to the team address
        uptownPanda.unlock(); // after liquidity is provided, tokens are unlocked
        wasPresaleEnded = true; // presale is ended so endPresale can't be called again
        isPresaleActive = false; // presale should not be active anymore
    }
}

// File: /home/keso/projects/crypto/uptown-panda-token/contracts/interfaces/IUniswapV2Helper.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.6.0;

interface IUniswapV2Helper {
    function sortTokens(address tokenA, address tokenB) external pure returns (address token0, address token1);

    function pairFor(address tokenA, address tokenB) external view returns (address pair);

    function getFactoryAddress() external view returns (address factory);

    function getRouterAddress() external view returns (address router);

    function getUniswapV2OracleAddress() external view returns (address oracle);
}

// File: /home/keso/projects/crypto/uptown-panda-token/contracts/interfaces/IUptownPanda.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.6.0;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface IUptownPanda is IERC20 {
    function mint(address _account, uint256 _amount) external;

    function isInitialized() external view returns (bool);

    function initialize(
        address _minter,
        address _weth,
        address _upFarm,
        address _upEthFarm,
        address _wethFarm,
        address _wbtcFarm
    ) external;

    function getMinter() external view returns (address);

    function unlock() external;

    function getListingPriceMultiplier() external view returns (uint256);

    function uniswapPair() external view returns (address);
}

// File: /home/keso/projects/crypto/uptown-panda-token/contracts/interfaces/IUptownPandaFarm.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.6.0;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface IUptownPandaFarm {
    function startFarming(
        address _upToken,
        address _farmToken,
        uint256 _initialFarmUpSupply
    ) external;

    function stake(uint256 _amount) external;

    function withdraw(uint256 _amount) external;

    function harvest() external;

    function claim() external;

    function harvestableReward() external view returns (uint256);

    function claimableHarvestedReward() external view returns (uint256);

    function totalHarvestedReward() external view returns (uint256);
}

// File: @openzeppelin/contracts/GSN/Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

import "../GSN/Context.sol";
/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor () internal {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(_owner == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}

// File: @openzeppelin/contracts/math/SafeMath.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");

        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return sub(a, b, "SafeMath: subtraction overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return div(a, b, "SafeMath: division by zero");
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return mod(a, b, "SafeMath: modulo by zero");
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts with custom message when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b != 0, errorMessage);
        return a % b;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: @uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router01.sol



pragma solidity >=0.6.2;

interface IUniswapV2Router01 {
    function factory() external pure returns (address);
    function WETH() external pure returns (address);

    function addLiquidity(
        address tokenA,
        address tokenB,
        uint amountADesired,
        uint amountBDesired,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB, uint liquidity);
    function addLiquidityETH(
        address token,
        uint amountTokenDesired,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external payable returns (uint amountToken, uint amountETH, uint liquidity);
    function removeLiquidity(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETH(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountToken, uint amountETH);
    function removeLiquidityWithPermit(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETHWithPermit(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountToken, uint amountETH);
    function swapExactTokensForTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapTokensForExactTokens(
        uint amountOut,
        uint amountInMax,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapExactETHForTokens(uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);
    function swapTokensForExactETH(uint amountOut, uint amountInMax, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapExactTokensForETH(uint amountIn, uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapETHForExactTokens(uint amountOut, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);

    function quote(uint amountA, uint reserveA, uint reserveB) external pure returns (uint amountB);
    function getAmountOut(uint amountIn, uint reserveIn, uint reserveOut) external pure returns (uint amountOut);
    function getAmountIn(uint amountOut, uint reserveIn, uint reserveOut) external pure returns (uint amountIn);
    function getAmountsOut(uint amountIn, address[] calldata path) external view returns (uint[] memory amounts);
    function getAmountsIn(uint amountOut, address[] calldata path) external view returns (uint[] memory amounts);
}

// File: @uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router02.sol



pragma solidity >=0.6.2;

import './IUniswapV2Router01.sol';

interface IUniswapV2Router02 is IUniswapV2Router01 {
    function removeLiquidityETHSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountETH);
    function removeLiquidityETHWithPermitSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountETH);

    function swapExactTokensForTokensSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
    function swapExactETHForTokensSupportingFeeOnTransferTokens(
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external payable;
    function swapExactTokensForETHSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
}